

# EXCEL <<<<<<<

By Bill Jelen

## Performing All Lookups in a Single Formula

The VLOOKUP function is one of Excel's most useful functions. It's likely that everyone in your accounting department can write VLOOKUP functions in their sleep. There's a different function supported by Excel gaining in popularity: LOOKUP. Excel Help indicates that LOOKUP is included in Excel for backwards compatibility with other spreadsheet programs and suggests that you should use the more modern VLOOKUP or HLOOKUP. More users are starting to use LOOKUP, however, because it allows you to perform a lookup on every value in a range in a single formula.

### The Two Kinds of VLOOKUP

If a VLOOKUP formula ends with FALSE as the optional fourth parameter, the formula looks for an exact match. If the VLOOKUP ends with TRUE or omits the fourth parameter, then it performs a range lookup. In the range lookup, the lookup table must be sorted ascending by the first column. VLOOKUP will return the value that is equal to or lower than the lookup value. Figure 1 shows a

**Figure 1**  
`=VLOOKUP(B2,$E$4:$F$10,2,TRUE)`

	A	B	C	D	E	F
1	Invoice	Amount	Bonus		Bonus Table	
2	1001	427	16			
3	1002	43	0		Amount	Bonus
4	1003	473	16		0	0
5	1004	99	0		100	3
6	1005	150	5		150	5
7	1006	225	7		200	7
8	1007	141	3		300	11
9	1008	500	25		400	16
10	1009	348	11		500	25
11	1010	705	25			
12	Total		108			

VLOOKUP using the range lookup. This version of VLOOKUP is great for replacing nested IF functions. The simple formula in C2 replaces the need to write a six-level nested IF formula:

```
=IF(B2>=500,25,IF(B2>=400,16,IF(B2>=300,11,IF(B2>=200,7,IF(B2>=150,5,IF(B2>=100,3,0))))))
```

Note that the LOOKUP function only does range lookups and won't do the exact match lookup.

### Contrasting LOOKUP and VLOOKUP

The syntax of LOOKUP is simpler than VLOOKUP. You specify a lookup value

and the lookup table. Excel always assumes that it is to return the value from the final column of the table. There's no need to specify TRUE or FALSE as the fourth parameter since LOOKUP always does a range lookup. If you wanted to change the worksheet in Figure 1 to use LOOKUP, the formula in C2 would be `=LOOKUP(B2,$E$4:$F$10)`.

An interesting side note about LOOKUP: If your lookup table is wider than it is tall, then LOOKUP

automatically switches to a horizontal lookup. It will look for the value in the top row of the table and return the result from the bottom row of the table.

### Performing Many Lookups at Once

LOOKUP clearly isn't as versatile as VLOOKUP, but there's one interesting trick that LOOKUP can do. In Figure 1, you have 10 lookup formulas and one sum formula—a total of 11 formulas to calculate the bonus. This is highly appropriate if you're sending out a month-end explanation of how the bonus is calculated. If your goal is to book a daily reserve

for accrued bonus, however, you might really only need a single formula that calculates the total bonus. Amazingly,

LOOKUP will allow you to specify an entire range of lookup values.

Instead of =LOOKUP(B2, you can specify =LOOKUP(B2:B11 as the lookup value. Provided that you hold down Ctrl+Shift while pressing Enter to accept the formula, Excel will perform the lookup on each individual cell in B2:B11.

Of course, this will return 10 values, and an Excel cell can only contain one value. To convert the 10 intermediate values, you need to wrap the formula in a function such as SUM. Figure 2 shows the formula and the result.

**Figure 2**

=SUM(LOOKUP(B2:B11,\$E\$4:\$F\$10))

	A	B	C	D	E	F	G	
1	Invoice	Amount			Bonus Table			
2	1001	427						
3	1002	43			Amount	Bonus		
4	1003	473			0	0		
5	1004	99			100	3		
6	1005	150			150	5		
7	1006	225			200	7		
8	1007	141			300	11		
9	1008	500			400	16		
10	1009	348			500	25		
11	1010	705						
12								
13	Bonus:	=SUM(LOOKUP(B2:B11,\$E\$4:\$F\$10))						
14		108						

## Calculating Additional Math with LOOKUP Results

The previous example assumed that a flat bonus amount was paid. What if you had to calculate the bonus as a sliding percentage of sales? In this case, you would want to multiply the result of the lookup with the sales amount in column

B. The SUMPRODUCT function will accept multiple ranges of numbers, multiply the corresponding values, and sum them. In Figure 3, the results of the multiple LOOKUPS are multiplied by the original sales amounts and summed.

Note that you don't have to press Ctrl+Shift+Enter when using SUMPRODUCT since that function is already expecting a series of arrays as arguments.

## Removing the Lookup Table

One final trick—and this would work with LOOKUP, VLOOKUP, or HLOOKUP. It's possible to remove the lookup table from the worksheet and place it directly inside your formula. Follow these steps:

1. In the cell containing your lookup formula, select the lookup table reference in the formula bar. Drag your mouse to highlight the characters \$E\$4:\$E\$10.

**Figure 3**

=SUMPRODUCT((LOOKUP(B2:B11,\$E\$4:\$F\$10)),(B2:B11))

	A	B	C	D	E	F	G	
1	Invoice	Amount			Bonus Table			
2	1001	427						
3	1002	43			Amount	Bonus		
4	1003	473			0	0.00%		
5	1004	99			100	3.00%		
6	1005	150			150	3.30%		
7	1006	225			200	3.50%		
8	1007	141			300	3.75%		
9	1008	500			400	4.00%		
10	1009	348			500	5.00%		
11	1010	705						
12								
13	Bonus:	=SUMPRODUCT((LOOKUP(B2:B11,\$E\$4:\$F\$10)),(B2:B11))						
14		126.355						

2. Press the F9 key to calculate that portion of the formula. Excel will change the reference to {0,0;100,0.03;150,0.033;200,0.035;300,0.0375;400,0.04;500,0.05}. In this array nomenclature, a comma indicates a new column, and a semicolon indicates a new row.
3. Press Enter to accept the formula with the lookup table hard-coded in the formula. You can now delete the original lookup table from the worksheet.

In this final example, the formula of =SUMPRODUCT((LOOKUP(B2:B11,{0,0;100,0.03;150,0.033;200,0.035;300,0.0375;400,0.04;500,0.05})),(B2:B11)) effectively replaces 27 cells worth of calculations with a single powerful formula. **SF**

*Bill Jelen is the host of MrExcel.com and the author of 25 books, including Excel Gurus Gone Wild. Send questions for future articles to IMA@MrExcel.com.*